
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.2.3.7.1 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2018-12_D19wk12_5G’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
8.2.3.7.1 
Test Group:
\ENDC\8_2
ATS Version:
Addition of EN-DC RRC test case 8.2.3.7.1 in FR1 path
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
Samsung Exynos Modem S5100, Qualcomm SDM855+SDX50
Verification Status:
PASS

4. Corrections required for ENDC TC 8.2.3.7.1
Note: The Change 4.3 from the 8.2.3.6.1 submission, R5s190155, is needed for this submission.
Change 4.1 

	Function name
	function fl_TC_8_2_3_7_1_TestBody()

	Reason for change
	SINR reporting is optional for UE based on its capability, this should be conditionally checked.

	Summary of change
	Changes are done to Report Config and Measurement Result contents based on this pc_ss_SINR_Meas .
Note: A draft Prose will be submitted in RAN5#83 meeting for this change

	TTCN module
	\8_2_3\RRCMeasurement_ENDC_NR.ttcn

	MCC160 Comment
	


Before Change
       function fl_TC_8_2_3_7_1_TestBody() runs on ENDC_NR_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryConditionFR1;

    var NR_AbsoluteCellPower_Type v_CellPower_LeavingConditionFR1;

    var RSRP_Range v_RsrpThres;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (present) MeasResults v_Expected_MeasResults;

    v_CellPower_EntryConditionFR1 := -79;

    v_CellPower_LeavingConditionFR1 := -91;

    v_CellPowerList_AtT0 := {

      cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

      cs_NR_CellPower(nr_Cell2, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_NR_CellPower(nr_Cell2, v_CellPower_EntryConditionFR1, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_NR_CellPower(nr_Cell2, v_CellPower_LeavingConditionFR1, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    //The SS re-adjusts the cell-specific reference signal level according to row "T0"

    fl_WaitForTrigger_And_ChangePower(v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message containing NR RRCReconfiguration message to setup NR measurement  and reporting of event A4.

    v_RsrpThres := 71;

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell2)};

    v_ReportConfigList := {cs_NR_ReportConfigEventA4(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_, ms1024)};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);  //@sic R5-192798 sic@

    f_NR_SendRRCReconfigurationContentsToEUTRA (omit, omit,v_MeasConfig);

    //@siclog "Step 3-5" siclog@

    //The SS re-adjusts the cell-specific reference signal level according to row "T1"

    //Check: Does the UE transmit ULInformationTransferMRDC message containing NR MeasurementReport message to report event A4 with the measured Results for NR Cell 2?

    //EXCEPTION: Step 5 below is repeated until 3 MeasurementReport messages are received from the UE

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell2);

    v_Expected_MeasResults := cr_38508_MeasResults_NeighCellsNR(tsc_MeasId1, v_PhysCellIdServing, {cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour)});

    fl_ChangePower_And_ReceiveMeasReports(v_CellPowerList_AtT1, v_Expected_MeasResults, "Steps 4-5", 4);

    //@siclog "Step 6" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T2"

    fl_WaitForTrigger_And_ChangePower(v_CellPowerList_AtT2);

    //@siclog "Step 7-8" siclog@

    //Running on EUTRA side

  } // End of fl_TC_8_2_3_7_1_TestBody
After Change
function fl_TC_8_2_3_7_1_TestBody() runs on ENDC_NR_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryConditionFR1;

    var NR_AbsoluteCellPower_Type v_CellPower_LeavingConditionFR1;

    var RSRP_Range v_RsrpThres;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (present) MeasResults v_Expected_MeasResults;

    //Cell Power levels according Table 8.2.3.7.1.3.2-1A - All values FFS

    v_CellPower_EntryConditionFR1 := -79;

    v_CellPower_LeavingConditionFR1 := -91;

    v_CellPowerList_AtT0 := {

      cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

      cs_NR_CellPower(nr_Cell2, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_NR_CellPower(nr_Cell2, v_CellPower_EntryConditionFR1, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_NR_CellPower(nr_Cell2, v_CellPower_LeavingConditionFR1, tsc_NR_NonSuitableOffCellSSS_EPRE)

    };

    //The SS re-adjusts the cell-specific reference signal level according to row "T0"

    fl_WaitForTrigger_And_ChangePower(v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message containing NR RRCReconfiguration message to setup NR measurement  and reporting of event A4.

    v_RsrpThres := 71;

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell2)};

    v_ReportConfigList := {cs_NR_ReportConfigEventA4(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_, ms1024)};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    if(not pc_ss_SINR_Meas) {

        v_ReportConfigList := {cs_NR_ReportConfigEventA4(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_,ms1024,cs_NR_MeasReportQuantity_RsrpRsrq)};

    } 
    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);  //@sic R5-192798 sic@

    f_NR_SendRRCReconfigurationContentsToEUTRA (omit, omit,v_MeasConfig);

    //@siclog "Step 3-5" siclog@

    //The SS re-adjusts the cell-specific reference signal level according to row "T1"

    //Check: Does the UE transmit ULInformationTransferMRDC message containing NR MeasurementReport message to report event A4 with the measured Results for NR Cell 2?

    //EXCEPTION: Step 5 below is repeated until 3 MeasurementReport messages are received from the UE

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell2);

    v_Expected_MeasResults := cr_38508_MeasResults_NeighCellsNR(tsc_MeasId1, v_PhysCellIdServing, {cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour)});

    if(not pc_ss_SINR_Meas) {

        v_Expected_MeasResults := cr_38508_MeasResults_NeighCellsNR(tsc_MeasId1, v_PhysCellIdServing, {cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour,cr_NR_MeasQuantityResults_RsrpRsrq)}, cr_NR_MeasQuantityResults_RsrpRsrq);

    }
    fl_ChangePower_And_ReceiveMeasReports(v_CellPowerList_AtT1, v_Expected_MeasResults, "Steps 4-5", 4);

    //@siclog "Step 6" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T2"

    fl_WaitForTrigger_And_ChangePower(v_CellPowerList_AtT2);

    //@siclog "Step 7-8" siclog@

    //Running on EUTRA side

  } // End of fl_TC_8_2_3_7_1_TestBody
5. Execution Log Files

5.1 Samsung Exynos Modem S5100
The Samsung Exynos Modem S5100 UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on EN-DC band combination DC_3A_n78A.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_2_3_7_1_Log.html
5.2 Qualcomm SDM855+SDX50

The Qualcomm SDM855+SDX50 UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on EN-DC band combination DC_1A_n78A.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_2_3_7_1_Log.html


(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

